package com.arrogames.util {
	
	/**
	 * This class represents an observable object, or "data" in the model-view paradigm. It can be subclassed to represent an object that the application wants to have observed.
	 *
	 *	<p>An observable object can have one or more observers. An observer may be any object that implements interface Observer. After an observable instance changes, an application calling the Observable's notifyObservers method causes all of its observers to be notified of the change by a call to their update method.</p>
	 *	<p>The order in which notifications will be delivered is unspecified. The default implementation provided in the Observerable class will notify Observers in the order in which they registered interest, but subclasses may change this order, use no guaranteed order, deliver notifications on separate threads, or may guarantee that their subclass follows this order, as they choose.</p>
	 *	<p>Note that this notification mechanism is has nothing to do with threads and is completely separate from the wait and notify mechanism of class Object.</p>
	 *	<p>When an observable object is newly created, its set of observers is empty. Two observers are considered the same if and only if the equals method returns true for them.</p>
	 * @author robinvdb
	 * 
	 */
	public class Observable {
		
		private var _observers:Array;
		private var _changed:Boolean;
		
		public function Observable() {
			_observers = [];
		}
		
		/**
		 * Adds an observer to the set of observers for this object, provided that it is not the same as some observer already in the set. The order in which notifications will be delivered to multiple observers is not specified. See the class comment.
		 * @param obs an observer to be added.
		 * 
		 */
		public function addObserver(obs:Observer):void {
			if (_observers.indexOf(obs) == -1) {
				_observers.push(obs);
			}
		}
		
		/**
		 * Deletes an observer from the set of observers of this object.
		 * @param obs the observer to be deleted.
		 * 
		 */
		public function deleteObserver(obs:Observer):void {
			for each (var o:Observer in _observers) {
				if (o == obs) {
					_observers.splice(o, 1);
					break;
				}
			}
		}
		
		/**
		 * Clears the observer list so that this object no longer has any observers. 
		 * 
		 */
		public function deleteObservers():void {
			_observers = [];
		}
		
		/**
		 * Returns the number of observers of this Observable object. 
		 * @return the number of observers of this object.
		 * 
		 */
		public function countObservers():int {
			return _observers.length;
		}
		
		/**
		 *	Marks this Observable object as having been changed; the hasChanged method will now return true. 
		 * 
		 */
		protected function setChanged():void {
			_changed = true;
		}
		
		/**
		 * Indicates that this object has no longer changed, or that it has already notified all of its observers of its most recent change, so that the hasChanged method will now return false. This method is called automatically by the notifyObservers methods. 
		 * 
		 */
		protected function clearChanged():void {
			_changed = false;
		}
		
		/**
		 * Tests if this object has changed. 
		 * @return true if and only if the setChanged method has been called more recently than the clearChanged method on this object; false otherwise.
		 * 
		 */
		public function hasChanged():Boolean {
			return _changed;
		}
		
		/**
		 *	If this object has changed, as indicated by the hasChanged method, then notify all of its observers and then call the clearChanged method to indicate that this object has no longer changed.
		 *	<p>Each observer has its update method called with two arguments: this observable object and null. In other words, this method is equivalent to: </p> 
		 * 	<p>notifyObservers(null)</p>
		 * 
		 */
		public function notifyObservers(obj:Object=null):void {
			if (hasChanged()) {
				clearChanged();
				for each (var o:Observer in _observers) {
					o.update(this, obj);
				}
			}
		}
		
	}
}